# Maintainer: Ronald van Haren <ronald.archlinux.org>
# Contributor: Ondrej Martinak <omartinak@gmail.com>

pkgname=libxdg-basedir
pkgver=1.2.0
pkgrel=1
pkgdesc="An implementation of the XDG Base Directory specifications."
arch=('i686' 'x86_64')
url="http://n.ethz.ch/student/nevillm/download/libxdg-basedir"
license=('MIT')
depends=('glibc')
source=(http://n.ethz.ch/student/nevillm/download/${pkgname}/${pkgname}-${pkgver}.tar.gz
	'LICENSE')
options=('!libtool')
sha1sums=('b924afe4e29c52358ad67476887e9ca0dcf90f7c'
          'c76efede60d0632d4241d718919d8b4ec62056b3')

build() {
   cd "${srcdir}/${pkgname}-${pkgver}"

   ./configure --prefix=/usr
   make
}

package() {
   cd "${srcdir}/${pkgname}-${pkgver}"
   make DESTDIR=${pkgdir} install

   # Install MIT license
   install -Dm644 ${srcdir}/LICENSE ${pkgdir}/usr/share/licenses/${pkgname}/LICENSE
}

